
Introduction

Physics, and in particular particle physics, has always made intensive use of computers.
Many things we see these days were either used first or invented by particle physicists.
The best known example is the World Wide Web. Also computer algebra has most
of its roots there because of the extreme complexity and size of the calculations in
theoretical particle physics. Today we will discuss an attempt to use recent techniques
in the field of artificial intelligence/game theory to solve very complicated systems of
equations that occur in the calculation of some reactions in particle physics.



Particle Physics

Particle physics is a part of physics that investigates the fundamental building blocks
of matter and the forces between them. These building blocks are:

quarks There are 6 types of quarks: u,d,s,c,b,t.

leptons Charged leptons are the electron, the muon and the tau. The chargeless
leptons are called neutrino’s and they come also in three varieties.

force mediators The photon (electromagnetism), the W± and Z (weak forces), the
gluon (strong interactions) and the graviton (gravity).

the Higgs particle A necessary ingredient to keep the best model we have (exclud-
ing gravity) physical (ie finite).

The quarks and the leptons come in two varieties: particles and anti-particles. The
force mediators and the Higgs particle are their own anti-particles. (With the W we
have a W+ and a W− which are each others anti-particles).



Currently there is no good theory that incorporates all interactions (forces), but we
have a very good theory that includes everything except for gravity. This is called
the ”standard model”. It predicted the Higgs particle although it could not predict
its mass.
When particles interact with each other, we call that a reaction. These interactions
can be seen as particles exchanging force mediators (or Higgs particles) as in:

Also a particle and an anti-particle can annihilate and form one or more force medi-
ators, or force mediators can produce a particle anti-particle pair.

The above pictures are called Feynman diagrams. In a proper theory, each element in
a diagram (lines and vertices) represents an element of a formula and when you want
to calculate a reaction, you have to write down all diagrams that can contribute to
it, write for each diagram its complete formula, square the sum of the diagrams, and
work out the formulas. This does involve quite some mathematics.



These formulas can become rather big in two ways:

• It can happen that one starts with one (complicated) diagram, both the input
and the output fit on a few lines, but at intermediate stages one could have many
Gbytes of formula.

• One may end up with a formula that takes millions of terms and this formula
needs to be integrated over by numerical means.

In both cases it should be clear that this is way beyond manual processing.



Computers!



FORM

For manipulating the formulas of the previous section (and many other types of for-
mulas) we use symbolic manipulation programs, also called computer algebra systems
(CAS).
The best known systems are Mathematica and Maple. They are designed to act a
bit like the way people work by hand and serve as a ”mathematicians work bench”.
Their generality however makes them pay dearly in efficiency. For most people this is
no objection.
If, however, one needs to do truely large calculations, it is better to use a different
package which is called FORM. This is designed for being fast and handling very big
formulas. The trade-off is that it uses a completely different programming model and
a number of complicated things are harder to program. Another advantage of FORM
is also that it is free and open source.
A few examples of (simple) FORM programs are:



Symbols a,b;

Local F = (a+b)^5;

Print;

.end

Time = 0.00 sec Generated terms = 6

F Terms in output = 6

Bytes used = 204

F =

b^5 + 5*a*b^4 + 10*a^2*b^3 + 10*a^3*b^2 + 5*a^4*b + a^5;



Symbols a,b,c,d,e,f,g;

Local F = (a+b+c+d+e+f+g)^30;

.end

Time = 6.51 sec Generated terms = 1947792

F Terms in output = 1947792

Bytes used = 99425612



Symbols n;

CFunction f;

Local Fibonacci = f(21);

Repeat;

id f(n?{>2}) = f(n-1)+f(n-2);

EndRepeat;

id f(1) = 1;

id f(2) = 1;

Print;

.end

Time = 0.05 sec Generated terms = 10946

Fibonacci Terms in output = 1

Bytes used = 20

Fibonacci =

10946;



The next example is more cryptic. It computes all possible drawings for the eighth
finals of the champions league in December 2012. It vetoes teams of the same nation-
ality playing against each other. There were two groups of eight teams and teams
from the first group had to play against teams of the second group. Each group had
two Spanish, one French and one Italian team. Other teams could not cause conflicts.

Tensor f;

Index i1,...,i8;

Local F = f(i1,...,i8)*e_(i1,...,i8)*e_(1,...,8);

Contract; * Generates the 8! permutations

id f(i1?{1,2},?a) = 0; * First Spanish team in group 2

id f(i1?,i2?{1,2},?a) = 0; * Other Spanish team in group 2

id f(i1?,i2?,3,?a) = 0; * Italian teams

id f(i1?,i2?,i3?,4,?a) = 0; * French teams

.end

Time = 0.05 sec Generated terms = 17088

F Terms in output = 17088

Bytes used = 527676



The program first generates all 8! = 40320 possibilities and then eliminates the ’for-
bidden’ combinations. There are 17088 possibilities left. Hence the chance that the
trial run and the real drawing would give the same result is 1/17088. Note how short
such a program can be if you know what you are doing.



The next example is one that shows the use of a (physics) library for calculating a
category of extremely difficult Feynman diagrams. The output does not show all the
steps, just a few crucial ones. The program:

#define LONGINT "1"

#include mincer3.h

Vectors Q,p1,...,p8;

Symbols n1,...,n9;

CFunction I;

Local F = I(5,5,5,5,5,5,5,5,5);

id I(n1?,...,n9?) = Q.p2^n9/<p1.p1^n1>/.../<p8.p8^n8>

*Q.Q^(-6+n1+...+n8-n9)*ep^3;

#call integral(no)

print;

.end

and the output (with selected intermediate statistics):



.

.

Time = 60.48 sec Generated terms = 204053

F Terms in output = 202857

noplane recursion Bytes used = 44371728

.

.

Time = 1160.61 sec Generated terms = 14082803

F Terms in output = 250615

Second recursion Bytes used = 69574612

.

.

Time = 1644.60 sec Generated terms = 19

F Terms in output = 19

Expression G scheme Bytes used = 1928

Time = 1644.60 sec Generated terms = 6

F Terms in output = 6

Bytes used = 308



F =

- 324095145340907149970425969992496467327106063029164883217475

/4844383464771741085138944

+ 458253263713157238125/96*ep^-3

+ 1148598445262358740134204625/33606144*ep^-2

- 46087171264173446923534618276861775/12241101400399872

*ep^-1

+ 271949513729659430423954995800000000*z5

- 17177818352514304675752734876850066875/96*z3;

We see here an example of a ‘simple’ answer with complicated intermediate results.

In the HEPGAME project we will be using FORM as the main tool for formula
manipulation. We will also use the fact that we have complete control over it to
extend it with new generic concepts that will facilitate our research.



Recursion Relations

Some categories of Feynman diagrams can be computed by the use of recursion re-
lations. One of the best known recursion relations in science is the Fibonacci series:

1,1,2,3,5,8,13,21,34,55,89,144,233,377,610,987,1597,2584,4181,

6765,10946,...

It is defined by

F (1) = 1

F (2) = 1

F (n) = F (n− 1) + F (n− 2)

We also call these difference equations. This would be a second order difference
equation, because it goes down to F (n−2). To solve an m-th order difference equation,
you need m startup values, and like with differential equations, some mathematics.
Sometimes such equations can be solved as in

F (n) =
(1 +
√

5)n − (1−
√

5)n

2n
√

5



Things can become much more complicated when there are more variables. Take for
instance the following diagram which represents a complicated integral:

I(a, b, c, d, e) =

a

b

c

d

e

For this integral one can write the recursion relation (never mind how):

(4−2ε−a−b−2e) I(a, b, c, d, e) = +aI(a+1, b, c, d, e−1)− aI(a+1, b, c−1, d, e)

+bI(a, b+1, c, d, e−1)− bI(a, b+1, c, d−1, e)

The variable ε is a parameter in the theory.
We see that either c, d or e is lowered by one while either a or b is raised by one.
If the variables c, d, e are positive integers, repeated application of this recursion
relation will make one of them zero eventually. As it turns out, if one of the parameters
a, b, c, d, e becomes zero or a negative integer, we have a closed form solution for this
integral. Hence, if either c, d, e or a, b, c are positive integers, we can solve the integral.



Such recursions are easy to program, but things can run out of hand at times:

Symbols a,b,c,d,e,ep,x1,x2;

CFunctions I,rat;

PolyRatfun rat;

Local expr = I(4,4,4,4,4);

repeat id I(a?pos_,b?pos_,c?pos_,d?pos_,e?pos_) =

rat(1,4-2*ep-a-b-2*e)*(

+a*I(a+1,b,c,d,e-1)-a*I(a+1,b,c-1,d,e)

+b*I(a,b+1,c,d,e-1)-b*I(a,b+1,c,d-1,e)

);

.end

Time = 6.57 sec Generated terms = 102070

expr Terms in output = 160

Bytes used = 51104



The next level of complication is that for the calculations we like to do one of the
parameters might not be an integer. If it is a or b, no problem. If it is c or d, also
no problem, because we flip the diagram. When e is not an integer, we do have a
problem.
In this case we can write a few more recursion relations. The basic relation is

(4−2ε−2a−b−e) I(a, b, c, d, e) = +bI(a−1, b+1, c, d, e)− bI(a, b+1, c, d, e)

+eI(a−1, b+1, c, d, e)− eI(a, b, c, d, e+1)

This relation can be used to bring the second argument down to one, by moving the
second term in the RHS to the left and the LHS to the right. Because of the coefficient
we cannot get it down to zero.
Due to symmetries we can bring also the first, third and fourth parameters down to
one this way.



The next steps we do in a Form program. First we define the two expressions

Symbols a,b,c,d,e,ep;

CFunction I;

Format NoSpaces;

Off Statistics;

Local expr1 = (4-2*ep-a-b-2*e)*I(a,b,c,d,e)

-a*I(a+1,b,c,d,e-1)+a*I(a+1,b,c-1,d,e)

-b*I(a,b+1,c,d,e-1)+b*I(a,b+1,c,d-1,e);

Local expr2 = (4-2*ep-2*a-b-e)*I(a,b,c,d,e)

-b*I(a-1,b+1,c,d,e)+b*I(a,b+1,c,d,e)

-e*I(a-1,b,c,d,e+1)+e*I(a,b,c-1,d,e+1);

.sort

We define two expressions that are supposed to be zero, because we have moved all
terms to the RHS. This makes it easier to combine them.



The expression to lower the first argument down to one is generated from the second
with

Local expr3 = expr2*replace_(a,b,b,a,c,d,d,c);

if ( expression(expr3) ) id I(a?,b?,c?,d?,e?) = I(b,a,d,c,e);

.sort

Here we flip the diagram upside down. This means that we exchange a ↔ b and
c ↔ d. In addition we have to exchange the corresponding positions in the function
that represents the integral.



After manually inspecting the equations we note that it is useful to make a new
expression that is a combination of the three expressions in which we shift e to e− 1
in the second and the third expressions. Moreover, we may assume now that a, b, c, d
are all one.

Local expr4 = expr1*replace_(a,1,b,1,c,1,d,1)

+expr2*replace_(a,1,b,1,c,1,d,1,e,e-1)

+expr3*replace_(a,1,b,1,c,1,d,1,e,e-1);

id I(1,0,1,1,e?) = I(0,1,1,1,e);

id I(1,1,0,1,e?) = I(0,1,1,1,e);

id I(1,1,1,0,e?) = I(0,1,1,1,e);

id I(2,0,1,1,e?) = I(0,2,1,1,e);

id I(1,2,1,0,e?) = I(2,1,0,1,e);

Bracket I;

Print +f expr4;

.end

We also apply some symmetries to the integrals that have now one zero and can be
evaluated.



The final answer of the program is

expr4=

+I(1,1,1,1,e)*(2-2*ep-2*e)

+I(1,1,1,1,-1+e)*(4-4*ep-2*e)

+I(0,2,1,1,-1+e)*(-2)

+I(2,1,0,1,e)*(2);

which is a relation that can be used to bring e in the range between 1 and 2. Usually
this will be 1 + ε.
At this point we have reduced the integrals to the single integral I(1, 1, 1, 1, 1 + ε).
This integral has to be done by different means and hence is called a master integral.
In the project we will run into very few master integrals (only two) and fortunately
those integrals are known.
The above shows basically what we will encounter also in the more complicated cases:
we obtain a number of relations and then we have to make various combinations
and transformations of these relations to reduce all integrals either to integrals of a
simpler type, or to a master integral. It may happen that the simpler integrals still
need similar techniques to be reduced to even simpler integrals, until we run into
integrals that we know how to do.



Games, Minimax and MCTS



Simplifying Equations

Often the result of symbolic computations is a lengthy formula that needs further
numerical processing. In the case of a Monte Carlo integral this may means many
thousands or even many millions of function evaluations. It is therefore important
that the output expression is written as economically as possible. This is called
simplification.

A very simple example occurs when one has an expression in a single variable as in

F = 5x5 + 12x4 + 4x3 − 12x2 − 23x + 7

= ((((5x + 12)x + 4)x− 12)x− 23)x + 7

In the first form there are 5 additions and at least 9 multiplications, while in the
second form there are 5 additions and 5 multiplications. Such a rewrite is called a
Horner scheme.



Things become messier when more than one variable is involved. In that case we
can use a Horner scheme in one of the variables and consider that the coefficients in
it are expressions in the remaining variables. Then these expressions are treated in
a similar way. While doing this, one may run into ‘common subexpressions’. In the
above example 12 would be such a subexpression. It pays to hunt those down and, if
they need computing, compute them only once.

The problem is that the efficiency of such a multivariate Horner scheme may de-
pend on the order in which the variables are treated. The differences can be quite
considerable.

The easy solution would be to try all orders, but once the number of variables
becomes large, this is not practical (20! = 2432902008176640000).

The various orderings of 20 variables define a tree. Selecting the first variable can
be done in 20 ways, the second in 19 ways, etc. The challenge is to find a path through
the tree that gives an optimal or near optimal solution.

